Math 2280-2
Generalized eigenvectors, Jordan canonical form, fundamental matrix solutions, and matrix exponentials

March 20, 2001

If the matrix A in the first order system of DE's given by dx/dt = Ax is diagonalizable, then an easy FMS solution to the system consists of an n by n matrix where each column is of the form exp(lambda*t)v, where the v's are a basis of eigenvectors of A. See today's notes from class. We can get exp(At) by then computing FMS(t)*inverse(FMS(0)). Here's the example we've been doing in class, **example 1 on page 354.**

```
> with(linalg):
> A:=matrix(2,2,[4,2,3,-1]);
```

$$A := \begin{bmatrix} 4 & 2 \\ 3 & -1 \end{bmatrix}$$

```
> eigenvects(A);
```

[-2, 1, {[1, -3]}], [5, 1, {[2, 1]}]

```
> FMS:=t->transpose(matrix(2,2,[exp(-2*t),-3*exp(-2*t),2*exp(5*t),ex
  p(5*t)]));
```

$$FMS := t \to \text{transpose}(\text{matrix}(2, 2, [\mathbf{e}^{(-2\,t)}, -3\,\mathbf{e}^{(-2\,t)}, 2\,\mathbf{e}^{(5\,t)}, \mathbf{e}^{(5\,t)}]))$$

```
> FMS(t);
    #our FMS, equation (14) page 354
```

$$\begin{bmatrix} \mathbf{e}^{(-2\,t)} & 2\,\mathbf{e}^{(5\,t)} \\ -3\,\mathbf{e}^{(-2\,t)} & \mathbf{e}^{(5\,t)} \end{bmatrix}$$

```
> evalm(FMS(t)&*inverse(FMS(0)));
    #and here is exp(At), example 5 page 359
```

$$\begin{bmatrix} \frac{1}{7}\mathbf{e}^{(-2\,t)}+\frac{6}{7}\mathbf{e}^{(5\,t)} & -\frac{2}{7}\mathbf{e}^{(-2\,t)}+\frac{2}{7}\mathbf{e}^{(5\,t)} \\ -\frac{3}{7}\mathbf{e}^{(-2\,t)}+\frac{3}{7}\mathbf{e}^{(5\,t)} & \frac{6}{7}\mathbf{e}^{(-2\,t)}+\frac{1}{7}\mathbf{e}^{(5\,t)} \end{bmatrix}$$

Of course, Maple can compute matrix exponentials (well, at least it can try).

```
> exponential(A,t);
     #this is a linalg command for exp(At)
```

$$\begin{bmatrix} \frac{1}{7}\mathbf{e}^{(-2\,t)}+\frac{6}{7}\mathbf{e}^{(5\,t)} & -\frac{2}{7}\mathbf{e}^{(-2\,t)}+\frac{2}{7}\mathbf{e}^{(5\,t)} \\ -\frac{3}{7}\mathbf{e}^{(-2\,t)}+\frac{3}{7}\mathbf{e}^{(5\,t)} & \frac{6}{7}\mathbf{e}^{(-2\,t)}+\frac{1}{7}\mathbf{e}^{(5\,t)} \end{bmatrix}$$

Now let's do a **bigger example. We use the matrix from #32 page 350:**

```
> A:=matrix([[11,-1,26,6,-3],
             [0,3,0,0,0],
             [-9,0,-24,-6,3],
             [3,0,9,5,-1],
             [-48,-3,-138,-30,18]]);
```

$$A := \begin{bmatrix} 11 & -1 & 26 & 6 & -3 \\ 0 & 3 & 0 & 0 & 0 \\ -9 & 0 & -24 & -6 & 3 \\ 3 & 0 & 9 & 5 & -1 \\ -48 & -3 & -138 & -30 & 18 \end{bmatrix}$$

```
> eigenvects(A);
```

$$[3, 3, \{[-1, 0, 1, 0, 6], [-1, 1, 0, 0, -3], [0, 0, 0, 1, 2]\}], [2, 2, \{[0, 0, -3, 1, -24], [1, 0, 0, 0, 3]\}]$$

This one was diagonalizable, even though it only has two different eigenvalues. So we could use the same method as in the first example, in order to find a FMS, and then exp(At). You might want to write an FMS in here:

We will use the diagonalization method of finding the matrix exponential. (See today's class notes.) We can diagonalize A, by using the matrix whose columns are a basis of eigenvectors:

```
> CT:=matrix([[1,0,0,0,3],[0,0,-3,1,-24],[-1,1,0,0,-3],[0,0,0,1,2],
              [-1,0,1,0,6]]);
```

$$CT := \begin{bmatrix} 1 & 0 & 0 & 0 & 3 \\ 0 & 0 & -3 & 1 & -24 \\ -1 & 1 & 0 & 0 & -3 \\ 0 & 0 & 0 & 1 & 2 \\ -1 & 0 & 1 & 0 & 6 \end{bmatrix}$$

```
> C:=transpose(CT);
     #columns are eigenvectors
```

$$C := \begin{bmatrix} 1 & 0 & -1 & 0 & -1 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & -3 & 0 & 0 & 1 \\ 0 & 1 & 0 & 1 & 0 \\ 3 & -24 & -3 & 2 & 6 \end{bmatrix}$$

```
> Diag:=evalm(inverse(C)&*A&*C);
    #this will be our diagonal matrix, as you recall from
    # linear algebra, and as we review today.
```

$$Diag := \begin{bmatrix} 2 & 0 & 0 & 0 & 0 \\ 0 & 2 & 0 & 0 & 0 \\ 0 & 0 & 3 & 0 & 0 \\ 0 & 0 & 0 & 3 & 0 \\ 0 & 0 & 0 & 0 & 3 \end{bmatrix}$$

```
> exponential(Diag,t);
    #matrix exponential of diagonal matrices is easy. (See class
  notes.)
```

$$\begin{bmatrix} \mathbf{e}^{(2\,t)} & 0 & 0 & 0 & 0 \\ 0 & \mathbf{e}^{(2\,t)} & 0 & 0 & 0 \\ 0 & 0 & \mathbf{e}^{(3\,t)} & 0 & 0 \\ 0 & 0 & 0 & \mathbf{e}^{(3\,t)} & 0 \\ 0 & 0 & 0 & 0 & \mathbf{e}^{(3\,t)} \end{bmatrix}$$

```
> matexpA:=t->evalm(C&*exponential(Diag,t)&*inverse(C));
     #this should be exp(At), see class notes
  matexpA(t);
```

$$matexpA := t \to \mathrm{evalm}(\text{`\&*`}(\text{`\&*`}(C, \mathrm{exponential}(Diag, t)), \mathrm{inverse}(C)))$$

$$[-8\,\mathbf{e}^{(2\,t)} + 9\,\mathbf{e}^{(3\,t)},\ \mathbf{e}^{(2\,t)} - \mathbf{e}^{(3\,t)},\ -26\,\mathbf{e}^{(2\,t)} + 26\,\mathbf{e}^{(3\,t)},\ -6\,\mathbf{e}^{(2\,t)} + 6\,\mathbf{e}^{(3\,t)},\ 3\,\mathbf{e}^{(2\,t)} - 3\,\mathbf{e}^{(3\,t)}]$$

$$[0,\ \mathbf{e}^{(3\,t)},\ 0,\ 0,\ 0]$$

$$[9\,\mathbf{e}^{(2\,t)} - 9\,\mathbf{e}^{(3\,t)},\ 0,\ 27\,\mathbf{e}^{(2\,t)} - 26\,\mathbf{e}^{(3\,t)},\ 6\,\mathbf{e}^{(2\,t)} - 6\,\mathbf{e}^{(3\,t)},\ -3\,\mathbf{e}^{(2\,t)} + 3\,\mathbf{e}^{(3\,t)}]$$

$$[-3\,\mathbf{e}^{(2\,t)} + 3\,\mathbf{e}^{(3\,t)},\ 0,\ -9\,\mathbf{e}^{(2\,t)} + 9\,\mathbf{e}^{(3\,t)},\ -2\,\mathbf{e}^{(2\,t)} + 3\,\mathbf{e}^{(3\,t)},\ \mathbf{e}^{(2\,t)} - \mathbf{e}^{(3\,t)}]$$

$$[48\,\mathbf{e}^{(2\,t)} - 48\,\mathbf{e}^{(3\,t)},\ 3\,\mathbf{e}^{(2\,t)} - 3\,\mathbf{e}^{(3\,t)},\ 138\,\mathbf{e}^{(2\,t)} - 138\,\mathbf{e}^{(3\,t)},\ 30\,\mathbf{e}^{(2\,t)} - 30\,\mathbf{e}^{(3\,t)},\ -15\,\mathbf{e}^{(2\,t)} + 16\,\mathbf{e}^{(3\,t)}]$$

```
> exponential(A,t);
```

```
    #check answer with Maple command
```

$$[-8\,e^{(2t)}+9\,e^{(3t)},\ e^{(2t)}-e^{(3t)},\ -26\,e^{(2t)}+26\,e^{(3t)},\ -6\,e^{(2t)}+6\,e^{(3t)},\ 3\,e^{(2t)}-3\,e^{(3t)}]$$

$$[0,\ e^{(3t)},\ 0,\ 0,\ 0]$$

$$[9\,e^{(2t)}-9\,e^{(3t)},\ 0,\ 27\,e^{(2t)}-26\,e^{(3t)},\ 6\,e^{(2t)}-6\,e^{(3t)},\ -3\,e^{(2t)}+3\,e^{(3t)}]$$

$$[-3\,e^{(2t)}+3\,e^{(3t)},\ 0,\ -9\,e^{(2t)}+9\,e^{(3t)},\ -2\,e^{(2t)}+3\,e^{(3t)},\ e^{(2t)}-e^{(3t)}]$$

$$[48\,e^{(2t)}-48\,e^{(3t)},\ 3\,e^{(2t)}-3\,e^{(3t)},\ 138\,e^{(2t)}-138\,e^{(3t)},\ 30\,e^{(2t)}-30\,e^{(3t)},\ -15\,e^{(2t)}+16\,e^{(3t)}]$$

What happens when A is not diagonalizable? This is where Jordan Canonical form enters the picture (see today's notes): We will use the matrix from **#31, page 350.**

```
> A:=matrix([[35,-12,4,30],[22,-8,3,19],[-10,3,0,-9],[-27,9,-3,-23]]
  );
```

$$A := \begin{bmatrix} 35 & -12 & 4 & 30 \\ 22 & -8 & 3 & 19 \\ -10 & 3 & 0 & -9 \\ -27 & 9 & -3 & -23 \end{bmatrix}$$

```
> eigenvects(A);
```

$$[1, 4, \{[0, 1, 3, 0], [1, 0, -1, -1]\}]$$

So the eigenspace is only 2-dimensional, even though lambda=1 has algebraic multiplicity 4. so the defect is 2.
We try to make chains: We use the matrix B=A-I since lambda=1.

```
> Id:=diag(1,1,1,1);
```

$$Id := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

```
> B:=evalm(A-Id);
```

$$B := \begin{bmatrix} 34 & -12 & 4 & 30 \\ 22 & -9 & 3 & 19 \\ -10 & 3 & -1 & -9 \\ -27 & 9 & -3 & -24 \end{bmatrix}$$

```
> evalm(B^2);
```

$$\begin{bmatrix} 42 & -18 & 6 & 36 \\ 7 & -3 & 1 & 6 \\ -21 & 9 & -3 & -18 \\ -42 & 18 & -6 & -36 \end{bmatrix}$$

```
> evalm(B^3);
```

$$\begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

Using the powers of B we have many choices for constructing our chains

```
> z:=vector([0,0,1,0]);
    #this will be the start of a chain of length
     #three
```

$$z := [0, 0, 1, 0]$$

```
> w:=evalm(B&*z);
```

$$w := [4, 3, -1, -3]$$

```
> v:=evalm(B&*w);
```

$$v := [6, 1, -3, -6]$$

```
> evalm(B&*v);
```

$$[0, 0, 0, 0]$$

```
>
```

So we have a chain of length three, namely {z,w,v}. Pick another lambda=1 eigenvalue (not v) and we will get a basis of R^4. The matrix of the linear transformation represented by A in standard coordinates will have Jordan normal form in the new basis:

```
> u:=vector([0,1,3,0]);
```

$$u := [0, 1, 3, 0]$$

```
> C:=augment(v,w,z,u);
    #enter chains in reverse order!
```

$$C := \begin{bmatrix} 6 & 4 & 0 & 0 \\ 1 & 3 & 0 & 1 \\ -3 & -1 & 1 & 3 \\ -6 & -3 & 0 & 0 \end{bmatrix}$$

```
> Jord:=evalm(inverse(C)&*A&*C);
```

$$Jord := \begin{bmatrix} 1 & 1 & 0 & 0 \\ 0 & 1 & 1 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

```
> jordan(A);
    #Maple also computes Jordan Canonical form. Of
    #course the blocks could be in any order
```

$$\begin{bmatrix} 1 & 1 & 0 & 0 \\ 0 & 1 & 1 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

```
> exponential(Jord,t);
    #exponentials of Jordan matrices are pretty easy to compute
    #from the power series definition, see class notes
```

$$\begin{bmatrix} \mathbf{e}^t & t\,\mathbf{e}^t & \frac{1}{2}t^2\,\mathbf{e}^t & 0 \\ 0 & \mathbf{e}^t & t\,\mathbf{e}^t & 0 \\ 0 & 0 & \mathbf{e}^t & 0 \\ 0 & 0 & 0 & \mathbf{e}^t \end{bmatrix}$$

```
> matexpA:=t->evalm(C&*exponential(Jord,t)&*inverse(C));
    #this should be the matrix exponential
  matexpA(t);
```

$$matexpA := t \to \mathrm{evalm}(`\&*`(`\&*`(C, \mathrm{exponential}(Jord, t)), \mathrm{inverse}(C)))$$

$$\begin{bmatrix} \mathbf{e}^t + 34\,t\,\mathbf{e}^t + 21\,t^2\,\mathbf{e}^t & -9\,t^2\,\mathbf{e}^t - 12\,t\,\mathbf{e}^t & 3\,t^2\,\mathbf{e}^t + 4\,t\,\mathbf{e}^t & 30\,t\,\mathbf{e}^t + 18\,t^2\,\mathbf{e}^t \\ 22\,t\,\mathbf{e}^t + \frac{7}{2}t^2\,\mathbf{e}^t & -\frac{3}{2}t^2\,\mathbf{e}^t - 9\,t\,\mathbf{e}^t + \mathbf{e}^t & \frac{1}{2}t^2\,\mathbf{e}^t + 3\,t\,\mathbf{e}^t & 19\,t\,\mathbf{e}^t + 3\,t^2\,\mathbf{e}^t \\ -10\,t\,\mathbf{e}^t - \frac{21}{2}t^2\,\mathbf{e}^t & \frac{9}{2}t^2\,\mathbf{e}^t + 3\,t\,\mathbf{e}^t & -\frac{3}{2}t^2\,\mathbf{e}^t - t\,\mathbf{e}^t + \mathbf{e}^t & -9\,t\,\mathbf{e}^t - 9\,t^2\,\mathbf{e}^t \\ -27\,t\,\mathbf{e}^t - 21\,t^2\,\mathbf{e}^t & 9\,t^2\,\mathbf{e}^t + 9\,t\,\mathbf{e}^t & -3\,t^2\,\mathbf{e}^t - 3\,t\,\mathbf{e}^t & \mathbf{e}^t - 24\,t\,\mathbf{e}^t - 18\,t^2\,\mathbf{e}^t \end{bmatrix}$$

```
> exponential(A,t);
    #check with Maple
```

$$\begin{bmatrix} e^t+34\,t\,e^t+21\,t^2\,e^t & -9\,t^2\,e^t-12\,t\,e^t & 3\,t^2\,e^t+4\,t\,e^t & 30\,t\,e^t+18\,t^2\,e^t \\ 22\,t\,e^t+\frac{7}{2}\,t^2\,e^t & -\frac{3}{2}\,t^2\,e^t-9\,t\,e^t+e^t & \frac{1}{2}\,t^2\,e^t+3\,t\,e^t & 19\,t\,e^t+3\,t^2\,e^t \\ -10\,t\,e^t-\frac{21}{2}\,t^2\,e^t & \frac{9}{2}\,t^2\,e^t+3\,t\,e^t & -\frac{3}{2}\,t^2\,e^t-t\,e^t+e^t & -9\,t\,e^t-9\,t^2\,e^t \\ -27\,t\,e^t-21\,t^2\,e^t & 9\,t^2\,e^t+9\,t\,e^t & -3\,t^2\,e^t-3\,t\,e^t & e^t-24\,t\,e^t-18\,t^2\,e^t \end{bmatrix}$$

```
>
```

What if we were using the chain method to find an FMS for the system in #31 page 350. Can use see why the columns of the following matrix would be a basis for our solution space?

```
> FMS:=t->evalm(C&*exponential(Jord,t));
  FMS(t);
```

$$FMS := t \rightarrow \text{evalm}(`\&*`(C, \text{exponential}(Jord, t)))$$

$$\begin{bmatrix} 6\,e^t & 6\,t\,e^t+4\,e^t & 3\,t^2\,e^t+4\,t\,e^t & 0 \\ e^t & t\,e^t+3\,e^t & \frac{1}{2}\,t^2\,e^t+3\,t\,e^t & e^t \\ -3\,e^t & -3\,t\,e^t-e^t & -\frac{3}{2}\,t^2\,e^t-t\,e^t+e^t & 3\,e^t \\ -6\,e^t & -6\,t\,e^t-3\,e^t & -3\,t^2\,e^t-3\,t\,e^t & 0 \end{bmatrix}$$

Finally, let's do **example 7, page 362:**

```
> A:=matrix(3,3,[3,4,5,0,5,4,0,0,3]);
```

$$A := \begin{bmatrix} 3 & 4 & 5 \\ 0 & 5 & 4 \\ 0 & 0 & 3 \end{bmatrix}$$

```
> exponential(A,t);
    #well, we may as well see what we're going to get!
```

$$\begin{bmatrix} e^{(3\,t)} & 2\,e^{(5\,t)}-2\,e^{(3\,t)} & 4\,e^{(5\,t)}-4\,e^{(3\,t)}-3\,t\,e^{(3\,t)} \\ 0 & e^{(5\,t)} & 2\,e^{(5\,t)}-2\,e^{(3\,t)} \\ 0 & 0 & e^{(3\,t)} \end{bmatrix}$$

```
> eigenvects(A);
```

$$[5, 1, \{[2, 1, 0]\}], [3, 2, \{[1, 0, 0]\}]$$

lambda = 3 is defective, so we will need to find a chain of length 2:

```
> iden:=matrix(3,3,[1,0,0,0,1,0,0,0,1]);
```

$$iden := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
> B:=evalm(A-3*iden);
```

$$B := \begin{bmatrix} 0 & 4 & 5 \\ 0 & 2 & 4 \\ 0 & 0 & 0 \end{bmatrix}$$

```
> nullspace(B^2);
    #will necessarily be 2-dim
```

$$\{[1, 0, 0], [0, -2, 1]\}$$

```
> v:=vector([0,-2,1]);
    #want a generalized eigenvector which is not an eigenvector
```

$$v := [0, -2, 1]$$

```
> u:=evalm(B&*v);
  evalm(B&*u);
```

$$u := [-3, 0, 0]$$

$$[0, 0, 0]$$

```
> w:=vector([2,1,0]);
    #lambda=5 eigenvector
```

$$w := [2, 1, 0]$$

```
> C:=augment(u,v,w);
```

$$C := \begin{bmatrix} -3 & 0 & 2 \\ 0 & -2 & 1 \\ 0 & 1 & 0 \end{bmatrix}$$

```
> Jord:=evalm(inverse(C)&*A&*C);
```

$$Jord := \begin{bmatrix} 3 & 1 & 0 \\ 0 & 3 & 0 \\ 0 & 0 & 5 \end{bmatrix}$$

```
> jordan(A);
    #just comparing
```

$$\begin{bmatrix} 5 & 0 & 0 \\ 0 & 3 & 1 \\ 0 & 0 & 3 \end{bmatrix}$$

```
> exponential(Jord,t);
```

$$\begin{bmatrix} e^{(3t)} & t\,e^{(3t)} & 0 \\ 0 & e^{(3t)} & 0 \\ 0 & 0 & e^{(5t)} \end{bmatrix}$$

```
> expAt:=t->evalm(C&*exponential(Jord,t)&*inverse(C));
  expAt(t);
```

$$expAt := t \to \mathrm{evalm}(`\&*`(`\&*`(C, \mathrm{exponential}(Jord, t)), \mathrm{inverse}(C)))$$

$$\begin{bmatrix} e^{(3t)} & -2\,e^{(3t)} + 2\,e^{(5t)} & -4\,e^{(3t)} - 3\,t\,e^{(3t)} + 4\,e^{(5t)} \\ 0 & e^{(5t)} & -2\,e^{(3t)} + 2\,e^{(5t)} \\ 0 & 0 & e^{(3t)} \end{bmatrix}$$

```
> FMS:=t->evalm(C&*exponential(Jord,t));
  FMS(t);
```

$$FMS := t \to \mathrm{evalm}(`\&*`(C, \mathrm{exponential}(Jord, t)))$$

$$\begin{bmatrix} -3\,e^{(3t)} & -3\,t\,e^{(3t)} & 2\,e^{(5t)} \\ 0 & -2\,e^{(3t)} & e^{(5t)} \\ 0 & e^{(3t)} & 0 \end{bmatrix}$$

Notice, we have done our operations in a different order from the text, and our FMS doesn't look exactly the same, but the matrix exponential is unique so that our answer agrees with the formual on page 363.

exp(A+B) equals exp(A)*exp(B) when A and B commute, but products might not be equal otherwise:

```
> A:=matrix(2,2,[4,2,3,-1]);
  B:=matrix(2,2,[3,0,0,3]);
  C:=matrix(2,2,[0,0,0,1]);
```

$$A := \begin{bmatrix} 4 & 2 \\ 3 & -1 \end{bmatrix}$$

$$B := \begin{bmatrix} 3 & 0 \\ 0 & 3 \end{bmatrix}$$

$$C := \begin{bmatrix} 0 & 0 \\ 0 & 1 \end{bmatrix}$$

```
> evalm(A&*B);
  evalm(B&*A);
    #matrices commute
```

```
exponential(A+B,t);
evalm(exponential(A,t)&*exponential(B,t));
  #exponential law holds
```

$$\begin{bmatrix} 12 & 6 \\ 9 & -3 \end{bmatrix}$$

$$\begin{bmatrix} 12 & 6 \\ 9 & -3 \end{bmatrix}$$

$$\begin{bmatrix} \frac{1}{7}\mathbf{e}^t + \frac{6}{7}\mathbf{e}^{(8\,t)} & \frac{2}{7}\mathbf{e}^{(8\,t)} - \frac{2}{7}\mathbf{e}^t \\ \frac{3}{7}\mathbf{e}^{(8\,t)} - \frac{3}{7}\mathbf{e}^t & \frac{6}{7}\mathbf{e}^t + \frac{1}{7}\mathbf{e}^{(8\,t)} \end{bmatrix}$$

$$\begin{bmatrix} \left(\frac{1}{7}\mathbf{e}^{(-2\,t)} + \frac{6}{7}\mathbf{e}^{(5\,t)}\right)\mathbf{e}^{(3\,t)} & \left(\frac{2}{7}\mathbf{e}^{(5\,t)} - \frac{2}{7}\mathbf{e}^{(-2\,t)}\right)\mathbf{e}^{(3\,t)} \\ \left(\frac{3}{7}\mathbf{e}^{(5\,t)} - \frac{3}{7}\mathbf{e}^{(-2\,t)}\right)\mathbf{e}^{(3\,t)} & \left(\frac{6}{7}\mathbf{e}^{(-2\,t)} + \frac{1}{7}\mathbf{e}^{(5\,t)}\right)\mathbf{e}^{(3\,t)} \end{bmatrix}$$

```
> evalm(A&*C);
  evalm(C&*A);
    #matrices don't commute
  exponential(A+C,t);
  evalm(exponential(A,t)&*exponential(C,t));
    #law of exponents may fail
>
```

$$\begin{bmatrix} 0 & 2 \\ 0 & -1 \end{bmatrix}$$

$$\begin{bmatrix} 0 & 0 \\ 3 & -1 \end{bmatrix}$$

$$\left[\frac{1}{2}\mathbf{e}^{((2+\sqrt{10})\,t)} - \frac{1}{10}\sqrt{10}\,\mathbf{e}^{(-(-2+\sqrt{10})\,t)} + \frac{1}{10}\sqrt{10}\,\mathbf{e}^{((2+\sqrt{10})\,t)} + \frac{1}{2}\mathbf{e}^{(-(-2+\sqrt{10})\,t)},\right.$$
$$\left. -\frac{1}{10}\sqrt{10}\,\mathbf{e}^{(-(-2+\sqrt{10})\,t)} + \frac{1}{10}\sqrt{10}\,\mathbf{e}^{((2+\sqrt{10})\,t)}\right]$$

$$\left[-\frac{3}{20}\sqrt{10}\,\mathbf{e}^{(-(-2+\sqrt{10})\,t)} + \frac{3}{20}\sqrt{10}\,\mathbf{e}^{((2+\sqrt{10})\,t)},\right.$$
$$\left.\frac{1}{2}\mathbf{e}^{((2+\sqrt{10})\,t)} + \frac{1}{10}\sqrt{10}\,\mathbf{e}^{(-(-2+\sqrt{10})\,t)} - \frac{1}{10}\sqrt{10}\,\mathbf{e}^{((2+\sqrt{10})\,t)} + \frac{1}{2}\mathbf{e}^{(-(-2+\sqrt{10})\,t)}\right]$$

$$\begin{bmatrix} \frac{1}{7}\mathbf{e}^{(-2\,t)}+\frac{6}{7}\mathbf{e}^{(5\,t)} & \left(\frac{2}{7}\mathbf{e}^{(5\,t)}-\frac{2}{7}\mathbf{e}^{(-2\,t)}\right)\mathbf{e}^{t} \\ \frac{3}{7}\mathbf{e}^{(5\,t)}-\frac{3}{7}\mathbf{e}^{(-2\,t)} & \left(\frac{6}{7}\mathbf{e}^{(-2\,t)}+\frac{1}{7}\mathbf{e}^{(5\,t)}\right)\mathbf{e}^{t} \end{bmatrix}$$

>